*Replication Instructions for

*Increasing Public Support for Transgender Rights through Superordinate LGBT+ Rights Framing: Evidence from a U.S. Survey Experiment on Transgender Participation in Sports

*Sam Whitt, Alixandra B. Yanus, Mark Setzler, Brian McDonald, Gordon Ballingrud, Tyler Coleman

*Below are instructions for replicating all manuscript and online appendix tables and figures in STATA using the dataset "PS transgender replication data.dta". Please contact Sam Whitt (swhitt@highpoint.edu) for questions regarding data replication. See also the dofile "PS transgender replication do file". 

*Note: You may need to install STATA packages for the catcibar, cibar, marginscontplot, and iebaltab commands. Use findit with the command name to identify and download the appropriate packets to install. 

*Note: In addition, some graphs require additional formatting using filename.grec files with the graph play command. To format a graph, simply run the command to generate the graph in the do file in STATA, then open the "Graph Editor" in STATA and click on the GREEN "Play Recording" button, then select "Browse" to select the grec file from the folder "grec files for STATA graph formatting" among Replication files. The name of the grec file is indicated in the note below the graph command in the do file for the specific graph you wish to format. This should automatically format the graph, which you may then save to a location of your choosing.

*Manuscript Replication

*"Stata user generated commands to install for replication purposes"

*"cohend"

ssc install cohend, replace

*"cibar"

ssc install cibar, replace

*"iebaltab from ietoolkit"

ssc install ietoolkit, replace

*"catcibar"

net install catcibar, from("https://aarondwolf.github.io/catcibar") replace

*"marginscontplot and marginscontplot2"

net install gr0056, from(http://www.stata-journal.com/software/sj13-3) replace
ssc install marginscontplot2, replace

*"regsensitivity"

ssc install regsensitivity, replace

*Replication in Text

*"A total of 1366 respondents completed our survey between May 19-25, 2021. Women represent 52% of the sample with few non-binary respondents (4 out of 1366). More people in our sample identify as Democrats (44%) than Republicans (32%) or Independents (20%), and our sample is predominantly White (78%), with 12% identifying as African-American and 12% as non-White Latino."

use "./PS transgender replication data.dta"
tab date
tab gender
tab partyid
tab race
tab dlatino

*"Out of 680 responses, 30.4% stated that transgender athletes should be able to participate in sports based on their self-identified gender, while 69.6% preferred that transgender athletes compete in sports based on their assigned gender at birth."

tab dtrans_sports if lgbt_first_txt==0

*"In Figure 2, we show that priming on LGBT+ rights led to a 9.1 percentage-point increase in overall support for transgender participation in sports based on self-identified gender (unpaired t-test=3.53, p<0.0002, Cohen's d=0.19)."
ttest dtrans_sports, by(lgbt_first_txt) unpaired unequal
cohend dtrans_sports lgbt_first_txt

*"The baseline treatment effect (20.1% increase relative to control) is strong and positive for people who report feeling very safe around LGBT+ people (49%) but declines by an average rate of 7.7% for people who report feeling only somewhat safe (34%), somewhat unsafe (11%), or very unsafe (6%) around LGBT+ people." (see also Table 1)

tab safe_lgbt

*"Democratic women are the most supportive of transgender inclusion in our study (47.6% in the control group), but our LGBT+ framing treatment group boosts support to 61% (unpaired t-test=2.34, p<0.01, Cohen's d=0.28)."

tab dtrans_sports if lgbt_first_txt==0 & female==1 & partyid==1
tab dtrans_sports if lgbt_first_txt==1 & female==1 & partyid==1
ttest dtrans_sports if female==1 & partyid==1, by(lgbt_first_txt) unpaired unequal
cohend dtrans_sports lgbt_first_txt if partyid==1 & female==1

*"Among Republican men and women, LGBT+ framing increases transgender support among Republican women from 13.5% to 24.2% (unpaired t-test=2.16, p<0.02, Cohen's d=0.28) and among Republican men from 16.4% to 26.3% (unpaired t-test=1.71, p<0.04, Cohen's d=0.25)."

tab dtrans_sports if lgbt_first_txt==0 & female==1 & partyid==2
tab dtrans_sports if lgbt_first_txt==1 & female==1 & partyid==2
ttest dtrans_sports if female==1 & partyid==2, by(lgbt_first_txt) unpaired unequal
cohend dtrans_sports lgbt_first_txt if partyid==2 & female==1

tab dtrans_sports if lgbt_first_txt==0 & female==0 & partyid==2
tab dtrans_sports if lgbt_first_txt==1 & female==0 & partyid==2
ttest dtrans_sports if female==0 & partyid==2, by(lgbt_first_txt) unpaired unequal
cohend dtrans_sports lgbt_first_txt if partyid==2 & female==0

*MS Table 1

reg dtrans_sports lgbt_first_txt, robust
reg dtrans_sports lgbt_first_txt revalphalgbtnational revalphalgbtstatelocal, robust
reg dtrans_sports lgbt_first_txt##c.safe_lgbt revalphalgbtnational revalphalgbtstatelocal i.partyid i.ideo i.gender age education hhincome devangelical ib2.race dlatino i.census_region rural if partyid<4  & ideo<6, robust

*MS Figure 2

cibar dtrans_sports, over(lgbt_first_txt) 
*Note additional formatting requires the "Figure 2 formatting.grec" file with the command graph play "Figure 2 formatting.grec" 

*Ms. Figure 3

cibar dtrans_sports if partyid<4, over1(lgbt_first_txt) over2(partyid) over3(female)
*Note additional formatting requires the "Figure 3 formatting.grec" file with the command graph play "Figure 3 formatting.grec" 


*Online Appendix Replication

*Appendix Table 1

sum dtrans_sports lgbt_first_txt  revlgbt_congress revlgbt_sc revalphalgbtnational revlgbt_state revlgbt_local revalphalgbtstatelocal safe_lgbt  i.gender age education i.partyid i.ideo i.race dlatino hhincome devangelical i.census_region rural

*Appendix Table 2

iebaltab revalphalgbtnational revalphalgbtstatelocal safe_lgbt partyid ideo gender age education hhincome devangelical race dlatino census_region rural, groupvar(lgbt_first_txt) savexls(balance)

*Appendix Figure 1

catcibar revlgbt_congress revlgbt_sc revlgbt_state revlgbt_local, over(lgbt_first_txt)
*Note additional formatting requires the "Appendix Figure 1 formatting.grec" file with the command graph play "Appendix Figure 1 formatting.grec" 

*Appendix Table 3a

factor revlgbt_congress revlgbt_sc
alpha revlgbt_congress revlgbt_sc

*Appendix Table 3b

factor revlgbt_state revlgbt_local
alpha revlgbt_state revlgbt_local

*Appendix Table 4

logit dtrans_sports lgbt_first_txt, robust
logit dtrans_sports lgbt_first_txt revalphalgbtnational revalphalgbtstatelocal, robust
logit dtrans_sports lgbt_first_txt revalphalgbtnational revalphalgbtstatelocal i.partyid i.ideo i.gender age education hhincome devangelical ib2.race dlatino i.census_region rural if partyid<4  & ideo<6, robust
logit dtrans_sports lgbt_first_txt##c.safe_lgbt revalphalgbtnational revalphalgbtstatelocal i.partyid i.ideo i.gender age education hhincome devangelical ib2.race dlatino i.census_region rural if partyid<4  & ideo<6, robust

*Appendix Figure 2

cibar dtrans_sports if ideo<6, over2(ideo) over1(lgbt_first_txt)
*Note additional formatting requires the "Appendix Figure 2 formatting.grec" file with the command graph play "Appendix Figure 2 formatting.grec" 

*Appendix Figure 3

reg dtrans_sports lgbt_first_txt##c.revalphalgbtnational, robust
marginscontplot  revalphalgbtnational lgbt_first_txt, at2(0 1) at1(1(1)4) ci
*Note additional formatting requires the "Appendix Figure 3 formatting.grec" file with the command graph play "Appendix Figure 3 formatting.grec" 

*Appendix Figure 4

reg dtrans_sports lgbt_first_txt##c.revalphalgbtstatelocal, robust
marginscontplot2  revalphalgbtstatelocal lgbt_first_txt, at2(0 1) at1(1(1)4) ci
*Note additional formatting requires the "Appendix Figure 4 formatting.grec" file with the command graph play "Appendix Figure 4 formatting.grec" 

*Power Calculations

power oneway, ngroups(2) n1(680) n2(683) power(0.80 0.90 0.95 0.99)

tab lgbt_first_txt partyid if partyid<4, chi2

anova dtrans_sports lgbt_first_txt##i.partyid if partyid<4

power twoway, n(1314) power(.80) nrows(2) ncols(3) varerror (0.21) factor(rowcol)

cohend dtrans_sports lgbt_first_txt if partyid==2

*Sensitivity Analysis

reg dtrans_sports lgbt_first_txt, robust

reg dtrans_sports lgbt_first_txt c.safe_lgbt revalphalgbtnational revalphalgbtstatelocal i.partyid i.ideo i.gender age education hhincome devangelical ib2.race dlatino i.census_re, robust

regsensitivity bounds dtrans_sports lgbt_first_txt c.safe_lgbt revalphalgbtnational revalphalgbtstatelocal i.partyid i.ideo i.gender age education hhincome devangelical ib2.race dlatino i.census_region rural if partyid<4  & ideo<6, plot

regsensitivity bounds dtrans_sports lgbt_first_txt c.safe_lgbt revalphalgbtnational revalphalgbtstatelocal i.partyid i.ideo i.gender age education hhincome devangelical ib2.race dlatino i.census_region rural if partyid<4  & ideo<6, oster 

log close